
8 Subroutines

In this lecture we will learn how to use subroutines, also known as functions, to modularize your code and make big programming tasks simpler to organize.

8.1 Defining a subroutine

Creating a subroutine is Perl is very easy.  You don't need to prototype or define your argument list.  This is a schematic on how to define a Perl subroutine.

sub subname {


...


some statements


...

}

That's it.  Let's try a simple example that you can type on your computer and play around with.

#!/usr/local/bin/perl -w

#

# Example 8.1 A simple subroutine example

#

# Here's the beginning of your main program.

#

print "I'm in the main program.\n";

&demo();

print "I'm back in the main program.\n";

exit;

#

# Here's our subroutine demonstration.

#

sub demo {


print "I'm in the subroutine now.  Weeee!\n";

}

This example will print out the following:

I'm in the main program.

I'm in the subroutine now. Weeee!

I'm back in the main program.

If you are new to programming, you may not understand what Perl is doing so let's dissect the code.  In this example you have one subroutine called demo and some other Perl statements that are not part of a subroutine.  When Perl starts executing the code, it makes one pass through the code and identifies the subroutines and the code that is not part of any subroutine.  I call the code not in subroutines main, after the convention we use for C programs to identify the code in the main function.  Then Perl executes the first statement that is not part of a subroutine.  In this case it's the line

print "I'm in the main program.\n";

Perl then continues with the line

&demo();

which tells Perl to execute the statements in the demo subroutine.  Perl then finds the demo subroutine and executes its only statement

print "I'm in the subroutine now.  Weeee!.\n";

Once the subroutine statement has been executed, Perl goes on to the next line in main

print "I'm back in the main program.\n";

Finally the code ends when Perl encounters the exit statement.

As you see above, the main program calls the demo subroutine with the syntax

&demo();

The ampersand, &, tells Perl that demo is a subroutine (just like $ identifies scalars, @ identifies arrays, and % identifies hashes).  The pair of parentheses is where you would put parameters if you wanted to pass data to demo.  Here we aren't sending any data to demo so the parentheses are empty.  The & and () are optional when calling subroutines as long as Perl can tell that you intend to call a subroutine.  The following are valid ways to call demo.

&demo();

&demo;

demo();

This way is not valid

demo;

because Perl can't tell if this is supposed to be a subroutine, a new built-in Perl operator, or a variable that is missing a $, @, or %.

8.2 Subroutine location

When you write subroutines into your code, you can put them all after the main code, or put all of them before so that your main code appears last.  There is no difference as far as Perl is concerned.  You can even mix your subroutines within your main code but I strongly advise you not to get into this habit as it make your code much harder to read.

You can also put your subroutines into separate files.  This can be particularly useful if you have many different Perl programs that do same operations over and over.  For example, one code you've written may have a particularly useful file opening subroutine that does all sorts of checks and corrections if it encounters errors.  When you write other Perl programs that open files, it would be better to use this subroutine again instead of rewriting a new one.  Sure, you can cut-and-paste the subroutine into your new programs, but if you need to change something, you have edit the subroutine in every file that it appears.  The best thing to do would be to put the subroutine into its own file and just tell your programs to get the subroutine from this one file if it's needed.  The way you can do this is to use the require keyword.  Here's a simple example.

Create a file called demo.pl with the following subroutine.

sub demo {


print "In demo subroutine in demo.pl file.\n";

}

return 1;

Now create a file called mainProgram.pl that contains this code.

#!/usr/bin/perl -w

#

require "demo.pl";

print "In main.\n";

demo();

print "Back in main.\n";

exit;

Run mainProgram.pl and you should see this output

In main.

In demo subroutine in demo.pl file.

Back in main.

What require is doing is that is looks for the demo.pl file and inserts it's contents where require appears.  However, before it actual does this, Perl reads the file demo.pl and looks for syntax errors, other require statements, and other things.  The 'return 1' at the end of demo.pl is necessary because require expects a true value.  If you leave it out, an undef is returned and require thinks there is a problem (such as an empty file).

8.3 Variable scope

So far we have only used global variables.  For small programs, global variables are ok but as your programs get longer, global variables become much harder to use.  Here is a trivial example to show what happens when you only use global variables.

#!/usr/bin/perl -w

#

# Example 8.3a Global variables.

# $i is global variable

#

$i = 1;

print "In main: \$i equals $i.\n";

number_1();

number_2();

print "In main: \$i equals $i.\n";

exit;

sub number_1() {


print "In subroutine 1: \$i equals $i.\n";


$i++;


print "In subroutine 1: \$i equals $i.\n";

}

sub number_2() {


print "In subroutine 2: \$i equals $i.\n";


$i++;


print "In subroutine 2: \$i equals $i.\n";

}

This will print

In main: $i equals 1.

In subroutine 1: $i equals 1.

In subroutine 1: $i equals 2.

In main: $i equals 2.

In subroutine 2: $i equals 2.

In subroutine 2: $i equals 3.

In main: $i equals 3.

If you look at the code you'll see that $i is assigned '1' in main.  When $i is first used in number_1(), it has the value '1' because we are using the same variable used in main.  Then when number() changes $i to be '2', we see that it is also '2' in main.  And it is also '2' in number_2() when Perl first enters number_2().  number_2() increments $i so that when Perl returns to main, $i is '3'.

This is ok if this is the behavior we wanted.  However, what if $i was a counter in main that counts one thing, and $i is also a counter in number_1() and number_2() that counts something else.  We wouldn't want main to mess up number_1()'s or number_2()'s counters and vice versa.

The way we can localize a variable, so that we can have two $i's that can hold different values is to use the my keyword.

#!/usr/bin/perl -w

#

# Example 8.3a Global variables.

# $i is global variable

#

my $i = 1;

print "In main: \$i equals $i.\n";

number_1();

print "In main: \$i equals $i.\n";

number_2();

print "In main: \$i equals $i.\n";

exit;

sub number_1() {


my $i = 2;


print "In subroutine 1: \$i equals $i.\n";


$i++;


print "In subroutine 1: \$i equals $i.\n";

}

sub number_2() {


my $i = 8;


print "In subroutine 2: \$i equals $i.\n";


$i++;


print "In subroutine 2: \$i equals $i.\n";

}

This will print

In main: $i equals 1.

In subroutine 1: $i equals 2.

In subroutine 1: $i equals 3.

In main: $i equals 1.

In subroutine 2: $i equals 8.

In subroutine 2: $i equals 9.

In main: $i equals 1.

Notice how by using my, $i in main was unaffected by the code in number_1() or number_2().  my says 'this is my variable and noone else can touch it'.  my ensures that the $i in main is localized to main, and the $i's in number_1() and number_2() are localized to those subroutines.  Besides main, you can localize variables using my anywhere there is a code block defined by two braces.  In this example we've seen how to localize $i between the braces starting and ending the subroutine number_1().  But we can also localize variables in other code structures such as for loops

# This $i local to the for loop only

for( $j=0; $j<10; $j++) {


my $i = 1;


....

}

and if-else blocks, just to name a few.

# These two $i's are local to each if-else block

if($j<10) {


my $i = 1;


....

}else{


my $i = 3;


....

}

my acts a a variable declaration so you use it on the variable either at the very beginning of the code or when you first use it.  It's sort of like C where you have declare a variable to be an int, char, float, etc. before you use it but in Perl we don't have to declare the variable type.

My recommendation is not to use global variables but to instead make every variable you use local by using the my keyword when ever possible.  You can also use my when declaring arrays and hashes.  Getting into this habit now will save you from headaches in the future.

Since you are going to use my from now on, you should also put 'use strict;' at the beginning of your programs.  What 'use strict' will do is that it will tell Perl that you intend to declare all of your variables with the my keyword.  Any variable that it finds that hasn't been declared with my will cause the program to halt.  It will do this because it thinks that the offending variable may be a typo.

Here's an example without my or use strict.

#!/usr/bin/perl -w

#

# $mydata is a typo.  We meant to put type $myData

# a second time but Perl doesn't know that.

# Perl will just create another variable called

# $mydata and assume that's what we intended.

#

$myData = "Here's some data";

...

some statements

...

$mydata = "Here's new data";

Now here's an example with the proper way to use my and use strict.

#!/usr/bin/perl -w

#

# $mydata is a typo.  Perl doesn't recognize

# $mydata because it hasn't been declared with my.

# Perl will stop with an error message.

#

use strict;

my $myData = "Here's some data";

....

some statements

....

$myData = "Here's new data"; # don't need 'my' again

Rule of thumb: Put in use strict at the beginning of your program and declare all variables with my.

8.4 Returning a value or values from a subroutine

A common task for subroutines is to calculate or retrieve some data and return it to the main program or another subroutine.  The way you tell a subroutine to return data is to use the return keyword.  Here's a little subroutine that will read a file and return it's contents to the main program as a single string.

sub readFile {


my $data = '';


open(FH, "<myData.txt") or

die "Can't open file: $!\n";


while(<FH>) {



$data .= $_;

}

close(FH);

return $data;

}

And here's how you could use the subroutine in a program

#!/usr/bin/perl -w

#

use strict;

my $fileContents = readFile();

print "The file contains $fileContents\n";

exit;

return can also take an array as an argument if you want to pass an array to your main program.  This code will do the same thing as the previous example except the contents of the file are passed as an array, one line per array element.

#!/usr/bin/perl -w

my @fileContents = readFile();

exit;

#

sub readFile {


my @data = ();


open(FH, "<myData.txt") or die "$!";


@data = <FH>;


close(FH);


return @data;

}

You can also use return to pass a hash back to the main program.  Demonstrating this is left as an exercise for the reader.

8.5 Passing data to a subroutine

We now know how to get data out of a subroutine but what if we want to give it data.  For example, in the readFile() subroutine above, it can only read from myData.txt.  It would be more convenient if we could give it any file name to open.  If you want to pass data, such as a file name to readFile(), just put that data between the parentheses.

#!/usr/bin/perl -w

#

my $myFile = "myData.txt";

my @fileContents = readFile( $myFile );

exit;

That's easy enough.  Now let's look at the readFile() code.

sub readFile {


my $file = shift @_;


open( FH, "<$file" ) or die "$!";


@data = <FH>;


close(FH);


return @data;

}

When you call a subroutine and put data between the parentheses, the data is put into a special array, @_, that is accessible by the subroutine.  In this case, the data in the variable $myFile is passed to the subroutine readFile( $myFile ).  The contents of $myFile, "myData.txt", are put into the array @_.  Within the subroutine, we get the name of the file by shifting @_ and assigning the first, and only, value to the variable $file.  Note that @_ is a default variable so you can leave it out of the shift statement.

sub readFile {


my $file = shift;  # same as shift @_;


...

If you have multiple scalars to pass to a subroutine, they all get put into @_ in the order that they appear in the argument list.

#

# This will print out "one two three"

#

&demo( "one", "two", "three" );

sub demo {


my $first = shift;


my $second = shift;


my $third = shift;


print "$first $second $third\n";

}

Using shift isn't the only way you can get the values out of @_.  The following will also work.

sub demo {


my ($first, $second, $third) = @_;


...

sub demo {


my $first = $_[0];


my $second = $_[1];


my $third = $_[2];


...

Finally you can also pass in an array and all of the array contents will be put into @_.

my @a = (1 .. 10);

&demo( @a );

#

sub demo {


my @b = @_;  # @b is equal to @a


...

}

8.6 Exercises

1.  Write a subroutine that will calculate the area of a rectangle.  This subroutine should take two arguments, height and width, and return the area.  Write code that accepts user input for the height and width, calculates the area with this subroutine, and prints the answer.

2. Modify the subroutine in problem 1 so that it calculates the area of a rectangle and the perimeter.  The perimeter is twice the sum of the height plus width.  The subroutine should accept height and width as arguments and return the area and perimeter in an array.  Write code that accepts user input for the height and width and prints out the area and perimeter of the rectangle.

